LFLFD - A FORTRAN IMPLEMENTATION OF THE FAST DECOUPLED LOAD FLOW TECHNIQUE

J.W. Bandler and W.M. Zuberek

SOS-82-8-L2

August 1983

LFLFD - A FORTRAN IMPLEMENTATION OF THE FAST DECOUPLED LOAD FLOW TECHNIQUE

J.W. Bandler and W.M. Zuberek

Abstract

This document contains a listing of the computer package LFLFD described in [1] for solving load flow problems with the use of the fast decoupled method. The package has been developed for the CDC 170/730 system with the NOS 1.4 level 552 operating system and the Fortran Extended (FTN) version 4.8 compiler. The listing contains a total of 325 lines (including 95 comments) constituting four subroutines. The listing does not include the Harwell package MA28 for solving sparse linear equations.

This work was supported by the Natural Sciences and Engineering Research Council of Canada under Grant G0647.

The authors are with the Simulation Optimization Systems Research Laboratory and the Department of Electrical and Computer Engineering, McMaster University, Hamilton, Canada L8S 4L7.

W.M. Zuberek is on leave from the Institute of Computer Science, Technical University of Warsaw, Warsaw, Poland. He is now with the Department of Electrical Engineering, Texas A&M University, College Station, TX 77843, U.S.A.

I. INTRODUCTION

LFLFD is a package of Fortran subroutines for solving load flow problems with the use of the fast decoupled method described by Stott and Alsac in [2]. This document contains a listing of the package LFLFD. The user's manual of the package presented together with illustrative examples is found in [1].

The package LFLFD has been developed for the CDC 170/730 system with the NOS 1.4 operating system and the Fortran Extended (FTN) version 4.8 compiler. The package is available at McMaster University in the form of a library of binary relocatable subroutines. The library is in the group indirect file LIBSPWR accessible under the charge RJWBAND.

The package LFLFD calls subroutines MA28A and MA28C of the Harwell Subroutine Library (Harwell package MA28) for solving sparse linear equations; the package MA28 must thus also be available when LFLFD is used. This document does not include MA28 package. Information concerning this package is found in [3].

The package LFLFD contains 325 lines of which 95 are comments.

II. REFERENCES

[1] J.W. Bandler and W.M. Zuberek, "LFLFD - A Fortran implementation of the fast decoupled load flow technique", Department of Electrical and Computer Engineering, McMaster University, Hamilton, Canada, Report SOS-82-8-U2, 1983.

[2] B. Stott and O. Alsac, "Fast decoupled load flow", IEEE Trans. Power Apparatus and Systems, vol. PAS-93, 1974, pp. 859-869.

[3] I.S. Duff, "MA28 - A set of Fortran subroutines for sparse unsymmetric linear equations", Computer Science and Systems Division, AERE Harwell, Oxfordshire, England, Report R.8730, 1980.

III. LISTING OF THE LFLFD PACKAGE

Subroutine	Number of lines (source text)	Number of words (compiled code)	Listing from page
LFLFD1	85	313	4
LFLFDA	141	666	5
LFLFDB	48	240	7
LFLFDC	51	221	8

```
      SUBROUTINE LFLFD1 (NB,NL,NZ,NR,INDR,INDC,IBT,Y,YS,V,S,W,LW,ITEL,     000001
     1                   VEPS,TIMEL,MODE,IFLAG)                            000002
      DIMENSION INDR(1),INDC(1),IBT(1),W(1)                                000003
      COMPLEX Y(1),YS(1),V(1),S(1)                                         000004
C                                                                          000005
C  THIS PACKAGE SOLVES LOAD FLOW PROBLEMS USING SPARSE MATRIX TECHNIQUES   000006
C  (HARWELL PACKAGE MA28) AND THE FAST DECOUPLED METHOD.                   000007
C                                                                          000008
C  LIBRARY : HARWELL PACKAGE MA28                                          000009
C                                                                          000010
C     NB     - NUMBER OF BUSES (EXCLUDING SLACK BUS),                      000011
C     NL     - NUMBER OF LOAD BUSES,                                       000012
C     NZ     - NUMBER OF NON-ZEROS IN BUS ADMITTANCE MATRIX,               000013
C     NR     - NUMBER OF NON-ZEROS IN LOAD BUS ADMITTANCE SUBMATRIX,       000014
C     INDR   - ROW INDEX OF SPARSE BUS ADMITTANCE MATRIX,                  000015
C     INDC   - COLUMN INDEX OF SPARSE BUS ADMITTANCE MATRIX,               000016
C     IBT    - VECTOR OF BUS TYPES (0 - LOAD BUS, 1 - GENERATOR BUS),      000017
C     Y      - SPARSE BUS ADMITTANCE MATRIX,                               000018
C     YS     - VECTOR OF SLACK BUS ADMITTANCES,                            000019
C     V      - COMPLEX BUS VOLTAGES (RECTANGULAR MODE),                    000020
C     S      - COMPLEX BUS INJECTED POWERS,                                000021
C     W      - REAL WORKSPACE,                                             000022
C     LW     - LENGTH OF THE WORSPACE W,                                   000023
C     ITEL   - LIMIT OF ITERATIONS,                                        000024
C     VEPS   - REQUIRED ACCURACY OF SOLUTION,                              000025
C     TIMEL  - LIMIT OF ITERATION TIME,                                    000026
C     MODE   - MODE OF OPERATION :                                         000027
C              0 - EVALUATE AND FACTORIZE APPROXIMATE JACOBIAN MATRICES,   000028
C              1 - PERFORM *P-DELTA* ITERATION (FOR FACTORIZED MATRICES),  000029
C              2 - PERFORM *Q-MOD.V* ITERATION (FOR FACTORIZED MATRICES),  000030
C              3 - PERFORM BOTH ITERATIONS (FOR FACTORIZED MATRICES),      000031
C     IFLAG  - RETURN FLAG :                                               000032
C             -2 - INCORRECT USE,                                          000033
C             -1 - INCORRECT PARAMETERS,                                   000034
C              0 - NORMAL RETURN (REQUIRED ACCURACY OBTAINED),             000035
C              1 - LIMIT OF ITERATIONS REACHED,                            000036
C              2 - LIMIT OF ITERATION TIME REACHED.                        000037
C                                                                          000038
      DATA NS1,NS2,NZ1,NZ2,IFL/0,0,0,0,-1/                                 000039
      IF(MODE.LT.0.OR.MODE.GT.3) GOTO 90                                   000040
      IF(MODE.EQ.0) GOTO 10                                                000041
      IFLAG=-2                                                             000042
      IF(NB.NE.NS1.OR.NL.NE.NS2.OR.NZ.NE.NZ1.OR.NR.NE.NZ2.OR.IFL.LT.0)     000043
     1 RETURN                                                              000044
      GOTO 50                                                              000045
   10 NBS=NB+1                                                             000046
      NS1=NB                                                               000047
      NS2=NL                                                               000048
      NZ1=NZ                                                               000049
      NZ2=NR                                                               000050
      LICN1=3*NZ1                                                          000051
      LICN2=3*NZ2                                                          000052
      LIRN1=NZ1+NS1+NS1                                                    000053
      LIRN2=NZ2+NS2+NS2                                                    000054
      JNRB=1                                                               000055
      JA1=JNRB+NBS                                                         000056
      JICN1=JA1+LICN1                                                      000057
      JIKP1=JICN1+LICN1                                                    000058
      JA2=JIKP1+5*NS1                                                      000059
      JICN2=JA2+LICN2                                                      000060
      JIKP2=JICN2+LICN2                                                    000061
      JRHS=JIKP2+5*NS2                                                     000062
      JW=JRHS+NS1                                                          000063
      JVM=JW+NS1                                                           000064
      JVA=JVM+NS1                                                          000065
```

```
      MXJ=JVA+NS1                                                       000066
      JIRN2=JRHS                                                        000067
      JIW2=JIRN2+LIRN2                                                  000068
      JW2=JIW2+8*NS2                                                    000069
      JIRN1=JA2                                                         000070
      JIW1=JIRN1+LIRN1                                                  000071
      JW1=JIW1+8*NS1                                                    000072
      MXJ=MAX0(MXJ,JW1+NS1,JW2+NS2)                                     000073
      J=LW-MXJ                                                          000074
      IF(J.LT.0) GOTO 90                                                000075
   50 CALL LFLFDA (NB,NL,NZ,NR,INDR,INDC,IBT,Y,YS,V,S,LICN1,LIRN1,      000076
     1              LICN2,LIRN2,W(JA1),W(JICN1),W(JIKP1),W(JA2),W(JICN2),  000077
     2              W(JIKP2),W(JRHS),W(JW),W(JVM),W(JVA),W(JNRB),        000078
     3              W(JIRN1),W(JIW1),W(JW1),W(JIRN2),W(JIW2),W(JW2),MODE,  000079
     4              ITEL,VEPS,TIMEL,IFLAG)                               000080
      IF(MODE.EQ.0)  IFL=IFLAG                                          000081
      RETURN                                                            000082
   90 IFLAG=-1                                                          000083
      RETURN                                                            000084
      END                                                               000085
C                                                                       000086
C                                                                       000087
      SUBROUTINE LFLFDA (NB,NLB,NZ,NZR,INDR,INDC,IBT,Y,YS,V,S,LICN1,    000088
     1                   LIRN1,LICN2,LIRN2,A1,ICN1,IKEEP1,A2,ICN2,      000089
     2                   IKEEP2,RHS,W,VM,VA,NRB,IRN1,IW1,W1,IRN2,IW2,   000090
     3                   W2,MODE,ITE,VEPS,TIMEX,IERR)                   000091
      DIMENSION INDR(1),INDC(1),IBT(1),ICN1(1),ICN2(1),IKEEP1(1),       000092
     1          IKEEP2(1),A1(1),A2(1),RHS(1),W(1),W1(1),W2(1),IW1(1),   000093
     2          IW2(1),NRB(1),VM(1),VA(1),IRN1(1),IRN2(1)               000094
      COMPLEX Y(1),YS(1),V(1),S(1)                                      000095
C                                                                       000096
C  THIS SUBROUTINE IMPLEMENTS THE FAST DECOUPLED ITERATIVE METHOD USING 000097
C  THE HARWELL PACKAGE *MA28* FOR SOLVING THE SPARSE SYSTEMS OF LINEAR  000098
C  EQUATIONS WITH REAL COEFFICIENTS.                                    000099
C                                                                       000100
C  REMARK : THE ARRAYS (RHS,W,VM,VA)                                    000101
C                AND   (IRN2,IW2,W2)                                    000102
C           AS WELL AS (A2,ICN2,IKEEP2,IRN2,IW2,W2)                     000103
C                AND   (IRN1,IW1,W1)                                    000104
C           SHARE THE SAME WORKSPACE.                                   000105
C                                                                       000106
      COMPLEX VV,CC,PW                                                  000107
      LOGICAL SWITCH                                                    000108
      CALL SECOND(TTIME1)                                               000109
      IERR=-2                                                           000110
      IF(MODE.NE.0) GOTO 2                                              000111
C                                                                       000112
C       SET ORDERING OF BUSES                                           000113
C                                                                       000114
      L=0                                                               000115
      K=NLB                                                             000116
      DO 4 I=1,NB                                                       000117
      IF(IBT(I).NE.0) GOTO 3                                            000118
      L=L+1                                                             000119
      NRB(I)=L                                                          000120
      GOTO 4                                                            000121
    3 K=K+1                                                             000122
      NRB(I)=K                                                          000123
    4 CONTINUE                                                          000124
C                                                                       000125
C       SET AND FACTORIZE SPARSE MATRICES OF REAL COEFFICIENTS          000126
C                                                                       000127
      NA1=0                                                             000128
      DO 6 I=1,NB                                                       000129
      NA1=NA1+1                                                         000130
```

```
      L=NA1                                                                    000131
      J1=1                                                                     000132
      IF(I.GT.1) J1=INDR(I-1)+1                                                000133
      J2=INDR(I)-1                                                             000134
      X=0.0                                                                    000135
      IF(AIMAG(YS(I)).NE.0.0) X=1.0/AIMAG(1.0/YS(I))                           000136
      DO 5 J=J1,J2                                                             000137
      IF (Y(J).EQ.(0.,0.)) GO TO 5                                             000138
      NA1=NA1+1                                                                000139
      KK=INDC(J)                                                               000140
      ICN1(NA1)=KK                                                             000141
      IRN1(NA1)=I                                                              000142
      XX=1.0/AIMAG(1.0/Y(J))                                                   000143
      X=X+XX                                                                   000144
      A1(NA1)=XX                                                               000145
    5 CONTINUE                                                                 000146
      ICN1(L)=I                                                                000147
      IRN1(L)=I                                                                000148
      A1(L)=-X                                                                 000149
    6 CONTINUE                                                                 000150
      U=0.1                                                                    000151
      CALL MA28A(NB,NA1,A1,LICN1,IRN1,LIRN1,ICN1,U,IKEEP1,IW1,W1,IFLAG)        000152
      IF(IFLAG.LT.0) RETURN                                                    000153
      L=0                                                                      000154
      DO 8 I=1,NB                                                              000155
      K=IBT(I)                                                                 000156
      IF(K.NE.0) GOTO 8                                                        000157
      LL=NRB(I)                                                                000158
      J1=1                                                                     000159
      IF(I.GT.1) J1=INDR(I-1)+1                                                000160
      J2=INDR(I)                                                               000161
      DO 7 J=J1,J2                                                             000162
      KK=INDC(J)                                                               000163
      IF(IBT(KK).NE.0) GOTO 7                                                  000164
      L=L+1                                                                    000165
      ICN2(L)=NRB(KK)                                                          000166
      IRN2(L)=LL                                                               000167
      A2(L)=-AIMAG(Y(J))                                                       000168
    7 CONTINUE                                                                 000169
    8 CONTINUE                                                                 000170
      U=0.1                                                                    000171
      CALL MA28A(NLB,NZR,A2,LICN2,IRN2,LIRN2,ICN2,U,IKEEP2,IW2,W2,IFLAG)       000172
      IF(IFLAG.LT.0) RETURN                                                    000173
C                                                                              000174
C       SET INITIAL VALUES AND CHECK LIMIT OF ITERATIONS                       000175
C                                                                              000176
    2 IT=0                                                                     000177
      CMX=0.0                                                                  000178
      CALL SECOND(TTIME2)                                                      000179
      IF(ITE.EQ.0) GOTO 90                                                     000180
C                                                                              000181
C       CONVERT VOLTAGES TO POLAR FORM                                         000182
C                                                                              000183
      DO 9 I=1,NB                                                              000184
      VV=V(I)                                                                  000185
      VM(I)=CABS(VV)                                                           000186
      VA(I)=ATAN2(AIMAG(VV),REAL(VV))                                          000187
    9 CONTINUE                                                                 000188
C                                                                              000189
C       ITERATION LOOP                                                         000190
C                                                                              000191
      CORRA=0.0                                                                000192
      CORRM=0.0                                                                000193
      IF(MODE.EQ.1) GOTO 10                                                    000194
      IF(MODE.EQ.2) GOTO 20                                                    000195
```

```
   10 IT=IT+1                                                              000196
      CALL LFLFDB(NB,NZ,Y,YS,V,VM,VA,S,INDR,INDC,IBT,NRB,A1,LICN1,ICN1,    000197
     1            IKEEP1,RHS,W,CORRA)                                      000198
      SWITCH=.TRUE.                                                        000199
      IF(IT.EQ.1.AND.(MODE.EQ.0.OR.MODE.EQ.3)) GOTO 20                     000200
      GOTO 40                                                              000201
   20 IT=IT+1                                                              000202
      CALL LFLFDC(NB,NLB,NZR,Y,YS,V,VM,VA,S,INDR,INDC,IBT,NRB,A2,LICN2,    000203
     1            ICN2,IKEEP2,RHS,W,CORRM)                                 000204
      SWITCH=.FALSE.                                                       000205
   40 CMX=AMAX1(CORRA,CORRM)                                               000206
      CALL SECOND(TTIME2)                                                  000207
      IF(CMX.LE.VEPS) GOTO 95                                              000208
      IF(TIMEX.LE.0.0) GOTO 44                                             000209
      IF(TTIME2-TTIME1.GE.TIMEX) GOTO 91                                   000210
   44 IF(ITE.LT.0) GOTO 45                                                 000211
      IF(IT.GE.ITE) GOTO 90                                                000212
   45 IF(CORRA.GT.2.0*CORRM) GOTO 10                                       000213
      IF(CORRM.GT.2.0*CORRA) GOTO 20                                       000214
      IF(SWITCH) GOTO 20                                                   000215
      GOTO 10                                                              000216
   90 IERR=1                                                               000217
      GOTO 96                                                              000218
   91 IERR=2                                                               000219
      GOTO 96                                                              000220
   95 IERR=0                                                               000221
   96 ITE=IT                                                               000222
      TIMEX=TTIME2-TTIME1                                                  000223
      VEPS=CMX                                                             000224
      RETURN                                                               000225
      END                                                                  000226
C                                                                          000227
C                                                                          000228
      SUBROUTINE LFLFDB (NB,NZ,Y,YS,V,VM,VA,S,INDR,INDC,IBT,NRB,A1,        000229
     1                   LICN1,ICN1,IKEEP1,RHS,W,CORRA)                    000230
      DIMENSION VM(1),VA(1),INDR(1),INDC(1),IBT(1),NRB(1),A1(1),           000231
     1          ICN1(1),IKEEP1(1),RHS(1),W(1)                              000232
      COMPLEX Y(1),YS(1),V(1),S(1)                                         000233
C                                                                          000234
C  THIS SUBROUTINE DETERMINES RIGHT-HAND-SIDES FOR ARGUMENT CORRECTIONS    000235
C  OF THE FAST DECOUPLED METHOD, SOLVES THE SPARSE SYSTEM OF LINEAR        000236
C  EQUATIONS AND UPDATES THE VOLTAGES.                                     000237
C                                                                          000238
      COMPLEX CURR,DELS,PW                                                 000239
      NBS=NB+1                                                             000240
C                                                                          000241
C       SET RIGHT-HAND-SIDES                                               000242
C                                                                          000243
      DO 20 I=1,NB                                                         000244
      J1=1                                                                 000245
      IF(I.GT.1) J1=INDR(I-1)+1                                            000246
      J2=INDR(I)                                                           000247
      CURR=YS(I)*V(NBS)                                                    000248
      DO 10 J=J1,J2                                                        000249
      K=INDC(J)                                                            000250
   10 CURR=CURR+Y(J)*V(K)                                                  000251
      PW=V(I)*CONJG(CURR)                                                  000252
      DELS=S(I)-PW                                                         000253
C     IF(IBT(I).NE.0) S(I)=CMPLX(REAL(S(I)),AIMAG(PW))                     000254
      RHS(I)=REAL(DELS)/VM(I)                                              000255
   20 CONTINUE                                                             000256
C                                                                          000257
C       SOLVE LINEAR EQUATIONS                                             000258
C                                                                          000259
      CALL MA28C(NB,A1,LICN1,ICN1,IKEEP1,RHS,W,1)                          000260
```

```
C                                                                       000261
C       UPDATE VOLTAGES                                                 000262
C                                                                       000263
      CORRA=0.0                                                         000264
      DO 50 I=1,NB                                                      000265
      V1=VM(I)                                                          000266
      DV=ABS(V1*SIN(RHS(I)))                                            000267
      V2=VA(I)+RHS(I)                                                   000268
      VA(I)=V2                                                          000269
      IF(DV.GT.CORRA) CORRA=DV                                          000270
      V(I)=CMPLX(V1*COS(V2),V1*SIN(V2))                                 000271
   50 CONTINUE                                                          000272
      RETURN                                                            000273
      END                                                               000274
C                                                                       000275
C                                                                       000276
      SUBROUTINE LFLFDC (NB,NLB,NZR,Y,YS,V,VM,VA,S,INDR,INDC,IBT,NRB,   000277
     1                   A2,LICN2,ICN2,IKEEP2,RHS,W,CORRM)              000278
      DIMENSION VM(1),VA(1),INDR(1),INDC(1),IBT(1),NRB(1),A2(1),        000279
     1          ICN2(1),IKEEP2(1),RHS(1),W(1)                           000280
      COMPLEX Y(1),YS(1),V(1),S(1)                                      000281
C                                                                       000282
C  THIS SUBROUTINE DETERMINES RIGHT-HAND-SIDES FOR MODULUS CORRECTIONS  000283
C  OF THE FAST DECOUPLED METHOD, SOLVES THE SPARSE SYSTEM OF LINEAR     000284
C  EQUATIONS AND UPDATES THE VOLTAGES.                                  000285
C                                                                       000286
      COMPLEX CURR,DELS,PW                                              000287
      NBS=NB+1                                                          000288
C                                                                       000289
C       SET RIGHT-HAND-SIDES                                            000290
C                                                                       000291
      DO 20 I=1,NB                                                      000292
      IF(IBT(I).NE.0) GOTO 20                                           000293
      L=NRB(I)                                                          000294
      J1=1                                                              000295
      IF(I.GT.1) J1=INDR(I-1)+1                                         000296
      J2=INDR(I)                                                        000297
      CURR=YS(I)*V(NBS)                                                 000298
      DO 10 J=J1,J2                                                     000299
      K=INDC(J)                                                         000300
   10 CURR=CURR+Y(J)*V(K)                                               000301
      PW=V(I)*CONJG(CURR)                                               000302
      DELS=S(I)-PW                                                      000303
      RHS(L)=AIMAG(DELS)/VM(I)                                          000304
   20 CONTINUE                                                          000305
C                                                                       000306
C       SOLVE LINEAR EQUATIONS                                          000307
C                                                                       000308
      CALL MA28C(NLB,A2,LICN2,ICN2,IKEEP2,RHS,W,1)                      000309
C                                                                       000310
C       UPDATE VOLTAGES                                                 000311
C                                                                       000312
      CORRM=0.0                                                         000313
      DO 50 I=1,NB                                                      000314
      IF(IBT(I).NE.0) GOTO 50                                           000315
      K=NRB(I)                                                          000316
      DV=ABS(RHS(K))                                                    000317
      IF(DV.GT.CORRM) CORRM=DV                                          000318
      V1=VM(I)                                                          000319
      V2=V1+RHS(K)                                                      000320
      VM(I)=V2                                                          000321
      V(I)=(V2/V1)*V(I)                                                 000322
   50 CONTINUE                                                          000323
      RETURN                                                            000324
      END                                                               000325
```